Appendix A

MONTECARLO:
! *A Monte Carlo study ensues.*
NAMES = g1 g2 g3 g4 p1 p2 p3 m1 m2 m3 m4 t1 t2 t3 t4 c1 c2 c3;
! *Desired name of each variable in the generated data.*
NOBS = 799;
! *Desired sample size for each dataset to be generated.*
NREPS = 10000;
! *Number of samples to be drawn. The number selected is consistent with recommendations for similar purposes (Muthén & Muthén, 2002). The number of samples drawn is analogous to sample size in most non-Monte Carlo studies.*
SEED = 82872;
! *Provides a starting place for the random draws. The seed was generated from a random numbers program. More than one seed can be used to check if the solution is stable. Two other random numbers, 36175 and 55771, were used as seeds for this purpose.*
GENERATE = g1 - c3 (2);
CATEGORICAL = g1 - c3;
! *Jointly specify that all of the variables to be generated will be scaled as an ordinal three-category rating scale (i.e., 2 thresholds per latent response variate).*
REPSAVE = ALL;
! *Save all raw datasets that are generated.*
SAVE = ces_ii_799*.DAT;
! *Name each saved data file ces_ii_799*; insert dataset number at *. Another file that lists each generated dataset, ces_ii_799list.dat, will automatically be created. This file may need to be altered to perform a subsequent external Monte Carlo study (to be discussed).*
RESULTS = ces_ii_799.RES;
! *Saves key output (e.g., parameter estimates) from each replication into a single datafile, which the author(s) named ces_ii_799. The exact (and order of) contents of the file, will be described at the end of the output file. This file can be read into an external program where various analyses (e.g., bias) can be performed.*

MODEL POPULATION:
! *The population model parameter values are provided below. Note that the population values are consistent with the relevant information in Table 1.*
gse BY g1@1 g2@1.028 g3@.914 g4@1.045 t4@.180;
pce BY p1@1 p2@1.425 p3@1.562;
me BY m1@1 m2@1.017 m3@1.075 m4@1.083;
te BY t1@1 t2@.938 t3@1.065 t4@.781 p1@.116;
cbe BY c1@1 c2@1.054 c3@1.065;
! *Latent variable names and unstandardized pattern coefficient values are provided above.*
gse@.624;
pce@.328;
me@.481;
te@.691;
cbe@.596;

! *The latent variable variance values are provided above.*

```
gse WITH pce@.245;
me WITH gse@.399;
me WITH pce@.220;
te WITH gse@.507;
te WITH pce@.251;
te WITH me@.322;
cbe WITH gse@.309;
cbe WITH pce@.168;
cbe WITH me@.393;
cbe WITH te@.269;
t1 WITH t3@.067;
c3 WITH g3@.099;
```

! *The latent variable covariance values are provided above.*

```
g1@.376;
g2@.340;
g3@.479;
g4@.320;
p1@.604;
p2@.334;
p3@.200;
m1@.519;
m2@.502;
m3@.444;
m4@.435;
t1@.309;
t2@.393;
t3@.216;
t4@.415;
c1@.404;
c2@.337;
c3@.324;
```

! *The measurement error variance values for latent response variates are provided above.*

```
[g1$1@-1.267 g1$2@0.311
g2$1@-1.108 g2$2@0.384
g3$1@-1.310 g3$2@0.311
g4$1@-1.108 g4$2@0.538
p1$1@-0.531 p1$2@0.520
p2$1@-0.638 p2$2@0.571
p3$1@-0.788 p3$2@0.722
m1$1@-0.905 m1$2@0.408
m2$1@-0.900 m2$2@0.793
m3$1@-0.738 m3$2@0.891
m4$1@-0.905 m4$2@0.608
t1$1@-1.274 t1$2@0.121
t2$1@-1.126 t2$2@0.301
```

```
        t3$1@-1.765 t3$2@-0.058
        t4$1@-1.036 t4$2@0.538
        c1$1@-1.303 c1$2@0.033
        c2$1@-1.485 c2$2@-0.140
        c3$1@-1.862 c3$2@-0.415];
```

! *The threshold values are provided above.*

MODEL COVERAGE:

! *Included so that the population values used to compute coverage equal the population values and not the start values to be specified under the MODEL command (to be discussed). Redundant commands are not annotated from this point forward.*

```
        gse BY g1@1 g2@1.028 g3@.914 g4@1.045;
        pce BY p1@1 p2@1.425 p3@1.562;
        me BY m1@1 m2@1.017 m3@1.075 m4@1.083;
        te BY t1@1 t2@.938 t3@1.065 t4@.781;
        cbe BY c1@1 c2@1.054 c3@1.065;
        gse@.624;
        pce@.328;
        me@.481;
        te@.691;
        cbe@.596;
        gse WITH pce@.245;
        me WITH gse@.399;
        me WITH pce@.220;
        te WITH gse@.507;
        te WITH pce@.251;
        te WITH me@.322;
        cbe WITH gse@.309;
        cbe WITH pce@.168;
        cbe WITH me@.393;
        cbe WITH te@.269;
        [g1$1@-1.267 g1$2@0.311
        g2$1@-1.108 g2$2@0.384
        g3$1@-1.310 g3$2@0.311
        g4$1@-1.108 g4$2@0.538
        p1$1@-0.531 p1$2@0.520
        p2$1@-0.638 p2$2@0.571
        p3$1@-0.788 p3$2@0.722
        m1$1@-0.905 m1$2@0.408
        m2$1@-0.900 m2$2@0.793
        m3$1@-0.738 m3$2@0.891
        m4$1@-0.905 m4$2@0.608
        t1$1@-1.274 t1$2@0.121
        t2$1@-1.126 t2$2@0.301
        t3$1@-1.765 t3$2@-0.058
        t4$1@-1.036 t4$2@0.538
```

```
        c1$1@-1.303 c1$2@0.033
        c2$1@-1.485 c2$2@-0.140
        c3$1@-1.862 c3$2@-0.415];

MODEL:
```
! *The theoretical model and start values for fitting it to each generated dataset are provided below. Note that the theoretical model differs from the population model consistent with Figure 1. The start values were taken from the results of fitting the theoretical model to the Myers et al.(2008) data in Step 2. The start values are given to provide a good starting point for estimation. Note that the start values are consistent with the relevant information in Table 1.*
```
        gse BY g1@1 g2*1.028 g3*.921 g4*1.045;
        pce BY p1@1 p2*1.218 p3*1.328;
        me BY m1@1 m2*1.018 m3*1.075 m4*1.084;
        te BY t1@1 t2*.901 t3*1.059 t4*.923;
        cbe BY c1@1 c2*1.054 c3*1.077;
        gse*.622;
        pce*.439;
        me*.481;
        te*.725;
        cbe*.592;
        gse WITH pce*.290;
        me WITH gse*.398;
        me WITH pce*.257;
        te WITH gse*.523;
        te WITH pce*.309;
        te WITH me*.333;
        cbe WITH gse*.315;
        cbe WITH pce*.196;
        cbe WITH me*.392;
        cbe WITH te*.276;
        cbe WITH pce*.208;
        [g1$1*-1.267 g1$2*0.311
        g2$1*-1.108 g2$2*0.384
        g3$1*-1.310 g3$2*0.311
        g4$1*-1.108 g4$2*0.538
        p1$1*-0.531 p1$2*0.520
        p2$1*-0.638 p2$2*0.571
        p3$1*-0.788 p3$2*0.722
        m1$1*-0.905 m1$2*0.408
        m2$1*-0.900 m2$2*0.793
        m3$1*-0.738 m3$2*0.891
        m4$1*-0.905 m4$2*0.608
        t1$1*-1.274 t1$2*0.121
        t2$1*-1.126 t2$2*0.301
        t3$1*-1.765 t3$2*-0.058
        t4$1*-1.036 t4$2*0.538
```

```
        c1$1*-1.303 c1$2*0.033
        c2$1*-1.485 c2$2*-0.140
        c3$1*-1.862 c3$2*-0.415];

OUTPUT:
          tech9;
```
! *Print error messages for each replication. This command helps identify problematic datasets.*

Appendix B

DATA:
! *Tells the program that existing dataset(s) will be analyzed.*
FILE = ces_ii_799_list.dat;
! *Directs program to a document that lists the datasets created previously (see text under SAVE = in Appendix A). The line for each problematic dataset (see text under tech 9 in Appendix A) was manually deleted prior to this run. Thus only the 9816 unproblematic datasets are used.*
TYPE = MONTECARLO;
! *An external Monte Carlo study ensues.*

VARIABLE:
NAMES = g1 g2 g3 g4 p1 p2 p3 m1 m2 m3 m4 t1 t2 t3 t4 c1 c2 c3;
! *The columns (i.e., variables) of each dataset are in the given order.*
USEVARIABLES = g1-c3;
! *Use all of the columns in the analysis. Subsequent commands are annotated in Appendix A.*
CATEGORICAL = g1-c3;

MODEL COVERAGE:
gse BY g1@1 g2@1.028 g3@.914 g4@1.045;
pce BY p1@1 p2@1.425 p3@1.562;
me BY m1@1 m2@1.017 m3@1.075 m4@1.083;
te BY t1@1 t2@.938 t3@1.065 t4@.781;
cbe BY c1@1 c2@1.054 c3@1.065;
gse@.624;
pce@.328;
me@.481;
te@.691;
cbe@.596;
gse WITH pce@.245;
me WITH gse@.399;
me WITH pce@.220;
te WITH gse@.507;
te WITH pce@.251;
te WITH me@.322;
cbe WITH gse@.309;
cbe WITH pce@.168;
cbe WITH me@.393;
cbe WITH te@.269;
[g1$1@-1.267 g1$2@0.311
g2$1@-1.108 g2$2@0.384
g3$1@-1.310 g3$2@0.311
g4$1@-1.108 g4$2@0.538
p1$1@-0.531 p1$2@0.520
p2$1@-0.638 p2$2@0.571
p3$1@-0.788 p3$2@0.722

```
     m1$1@-0.905 m1$2@0.408
     m2$1@-0.900 m2$2@0.793
     m3$1@-0.738 m3$2@0.891
     m4$1@-0.905 m4$2@0.608
     t1$1@-1.274 t1$2@0.121
     t2$1@-1.126 t2$2@0.301
     t3$1@-1.765 t3$2@-0.058
     t4$1@-1.036 t4$2@0.538
     c1$1@-1.303 c1$2@0.033
     c2$1@-1.485 c2$2@-0.140
     c3$1@-1.862 c3$2@-0.415];

MODEL:
     gse BY g1@1 g2*1.028 g3*.921 g4*1.045;
     ce BY p1@1 p2*1.218 p3*1.328;
     me BY m1@1 m2*1.018 m3*1.075 m4*1.084;
     te BY t1@1 t2*.901 t3*1.059 t4*.923;
     cbe BY c1@1 c2*1.054 c3*1.077;
     gse*.622;
     pce*.439;
     me*.481;
     te*.725;
     cbe*.592;
     gse WITH pce*.290;
     me WITH gse*.398;
     me WITH pce*.257;
     te WITH gse*.523;
     te WITH pce*.309;
     te WITH me*.333;
     cbe WITH gse*.315;
     cbe WITH pce*.196;
     cbe WITH me*.392;
     cbe WITH te*.276;
      [g1$1*-1.267 g1$2*0.311
      g2$1*-1.108 g2$2*0.384
      g3$1*-1.310 g3$2*0.311
      g4$1*-1.108 g4$2*0.538
      p1$1*-0.531 p1$2*0.520
      p2$1*-0.638 p2$2*0.571
      p3$1*-0.788 p3$2*0.722
      m1$1*-0.905 m1$2*0.408
      m2$1*-0.900 m2$2*0.793
      m3$1*-0.738 m3$2*0.891
      m4$1*-0.905 m4$2*0.608
      t1$1*-1.274 t1$2*0.121
```

```
    t2$1*-1.126 t2$2*0.301
    t3$1*-1.765 t3$2*-0.058
    t4$1*-1.036 t4$2*0.538
    c1$1*-1.303 c1$2*0.033
    c2$1*-1.485 c2$2*-0.140
    c3$1*-1.862 c3$2*-0.415];

SAVEDATA:
    RESULTS = ces_ii_799.RES;

OUTPUT:
    tech9;
```